```
> #proof of propery 1.2
restart;
with(linalg):

#Taylor expansion of 2D time-dependent vectro field
#V(x,y,z) = (uf(x,y,z) , vf(x,y,z) )^T
#z is the time component

uf := u + u_x*x + u_y*y + u_z*z
+ u_xx*x*x + u_xy*x*y + u_xz*x*z
+ u_yy*y*y + u_yz*y*z + u_zz*z*z:

vf := v + v_x*x + v_y*y + v_z*z
+ v_xx*x*x + v_xy*x*y + v_xz*x*z
+ v_yy*y*y + v_yz*y*z + v_zz*z*z:

#feature line passes through (0,0,0)
u := 0:
v := 0:

#partials of V
uf_x := diff(uf,x):
uf_y := diff(uf,y):
uf_z := diff(uf,z):

vf_x := diff(vf,x):
vf_y := diff(vf,y):
vf_z := diff(vf,z):


#the feature flow field F = (uff,vff,wff)^T
uff := uf_y*vf_z - vf_y*uf_z:
vff := uf_z*vf_x - vf_z*uf_x:
wff := uf_x*vf_y - vf_x*uf_y:

#gradient field G

#HG := (det(V,V_x) , det(V,V_y), det(V,V_z))^T
hugf := vf*uf_x -uf*vf_x:
hvgf := vf*uf_y -uf*vf_y:
hwgf := vf*uf_z -uf*vf_z:

#Correction Field G = (ugf,vgf,wgf)^T = -F/|F| \times HG
ugf := (hvgf*wff - hwgf*vff)/(uff^2 + vff^2 + wff^2)^(1/2):
vgf := (hwgf*uff - hugf*wff)/(uff^2 + vff^2 + wff^2)^(1/2):
wgf := (hugf*vff - hvgf*uff)/(uff^2 + vff^2 + wff^2)^(1/2):

#partials of G
ugf_x := diff(ugf,x):
vgf_x := diff(vgf,x):
wgf_x := diff(wgf,x):

ugf_y := diff(ugf,y):
vgf_y := diff(vgf,y):
wgf_y := diff(wgf,y):

ugf_z := diff(ugf,z):
vgf_z := diff(vgf,z):
wgf_z := diff(wgf,z):

#we are interested in the behavior at (0,0,0),
```

```
#all necessary derivatives are computed
x := 0:
y := 0:
z := 0:
```

Warning, the protected names norm and trace have been redefined and unprotected

```
> #check some values
factor(uff);
factor(vff);
factor(wff);
```

$$u\_y\, v\_z - v\_y\, u\_z$$

$$u\_z\, v\_x - v\_z\, u\_x$$

$$u\_x\, v\_y - v\_x\, u\_y$$

```
>
factor(ugf);
factor(vgf);
factor(wgf);
```

$$0$$

$$0$$

$$0$$

```
> #nabla G
NG := Matrix([
[factor(ugf_x), factor(ugf_y), factor(ugf_z)],
[factor(vgf_x), factor(vgf_y), factor(vgf_z)],
[factor(wgf_x), factor(wgf_y), factor(wgf_z)]
]):
```

```
> #absF = |F|
absF := factor(sqrt(uff^2 + vff^2 + wff^2));
```

$$absF := (u\_y^2\, v\_z^2 - 2\, u\_y\, v\_z\, v\_y\, u\_z + v\_y^2\, u\_z^2 + u\_z^2\, v\_x^2 - 2\, u\_z\, v\_x\, v\_z\, u\_x + v\_$$

$$- 2\, u\_x\, v\_y\, v\_x\, u\_y + v\_x^2\, u\_y^2)^{\left(\frac{1}{2}\right)}$$

```
> #compute eigenvectros and eigenvalues to prove the property
eigenvectors(NG);
```

$$\left[0, 1, \left\{\left[\frac{u\_y\, v\_z - v\_y\, u\_z}{u\_x\, v\_y - v\_x\, u\_y}, -\frac{-u\_z\, v\_x + v\_z\, u\_x}{u\_x\, v\_y - v\_x\, u\_y}, 1\right]\right\}\right], \Big[-(u\_y^2\, v\_z^2 - 2\, u\_y\, v\_z\, v\_y\, u\_$$

$$- 2\, u\_z\, v\_x\, v\_z\, u\_x + v\_z^2\, u\_x^2 + u\_x^2\, v\_y^2 - 2\, u\_x\, v\_y\, v\_x\, u\_y + v\_x^2\, u\_y^2)^{\left(\frac{1}{2}\right)}, 2,$$

$$\left\{\left[1, 0, \frac{v\_y\, u\_z - u\_y\, v\_z}{u\_x\, v\_y - v\_x\, u\_y}\right], \left[0, 1, \frac{-u\_z\, v\_x + v\_z\, u\_x}{u\_x\, v\_y - v\_x\, u\_y}\right]\right\}\Bigg]$$

>